
8.  Using Excel for Aquifer Test Analysis

8.1 Introduction

Many of the aquifer test analysis methods were developed in an era when computers were 
not widely available, and graphical matching was the only way to fit type curves to drawdown 
data. Although graphical matching works very well for simple aquifer models (such as the Theis 
solution), when the number of parameters exceeds 3 or 4, graphical matching becomes 
complicated and time consuming. This limitation is readily overcome today by using computers 
to analyze aquifer tests. We have already discussed the use of the program PEST for parameter 
estimation. In this lecture, we take another approach--using Microsoft Excel. A key advantage of 
using Excel is the ability to create an interactive environment that not only facilitates curve 
matching but also helps you better understand the aquifer response to pumping.

8.2 Creating a Scroll Bar in an Excel Worksheet

The key to creating an interactive environment is the ability to change an aquifer 
parameter (for example, transmissivity) and to view the effect of this change on the type 
curve(s). Although it is possible to use a worksheet cell to hold a parameter and then repeatedly 
enter new values into the cell, this approach is still too tedious. Instead, we will use a “scroll bar” 
to vary parameter values.

A scroll bar is created using the “Control Toolbox”. This toolbox is shown in Fig. 8.1. 
(Note: The Control Toolbox in your Excel program might have a different arrangement. Also, it 
might be “docked” to the edge of the worksheet, or “floating” on the worksheet.) If this toolbox 
is not visible, you can display it by clicking the “View” menu, then selecting “Toolbars”, and 
then selecting “Control Toolbox”.

Figure 8.1. The Control Toolbox in Excel.

To create a scroll bar on a worksheet, click the Scroll Bar button on the Control Toolbox, 
then draw (depress left mouse button and drag) a long skinny rectangle on the worksheet to 
outline the scrollbar. When you are done, your worksheet should look something like Figure 8.2. 
Note that (1) the small white squares around the scrollbar allows you to reshape the scrollbar, 
and (2) the Control Mode button in the Control Toolbox is now depressed, indicating that you 
can change the properties of the scroll bar. (Note that the scroll bar need not be aligned with the 
worksheet cells.)

Scroll Bar buttonControl Mode button

Figure 8.2. Creating a scrollbar on an Excel worksheet.

Next, right click on the scroll bar. A pop-up menu appears. Select “Properties”. This 
displays the Properties window for the scroll bar (Figure 8.3). In the Properties window, enter a 
cell to be linked to the scroll bar. In our example, we will use cell H3 as the linked cell. Also, 
change the “Max” value to 10,000. 

To return the worksheet to its normal operating mode, click the Control Mode button so it 
is no longer depressed. Test the scroll bar by dragging the slider. You should see the value in cell 
H3 change as you drag the slider. When the slider is at the left end of the scrollbar, the value in 
cell H3 should be 0. When the slider is at the right end of the scrollbar, the value in cell H3 
should be 10000. Note: If you want to modify the toolbar at a later time, just click the Control 
Mode button and then click on the toolbar. You may move it somewhere else, change the linked 
cell, etc.

Next, we convert the value in cell H3 into a transmissivity value, which we will put in 
cell F3. Suppose we want the transmissivity value to vary from 10-10 to 100. The formula for cell 
F3 is =10^((H3/1000)-10). Now if you drag the slider, the value in cell F3 should change from 
10-10 to 100. (See Figure 8.4.)

Our final step is to repeat the above procedure to create another scroll bar to change the 
storage coefficient. The storage coefficient value will be put in cell L3.

“Control Mode” button (depressed)

scrollbar

linked cell (H3)

slider

Figure 8.3. The Properties window showing the properties of a scroll bar.

Figure 8.4. Using a scrollbar to change transmissivity value in cell F3.

Change the Max value to 10000

Enter the linked cell here. (H3 in our example)

8.3 Creating an Interactive Theis Type Curve

We now create a Theis type curve that can be controlled by the transmissivity and storage 
coefficient scroll bars. First, we use Visual Basic to create a user-defined function to evaluate the 
exponential integral E1(u), using the formulas given by (3.15) and (3.16). Before we create the 
function, we first save the workbook as notes09_Theis.xls. The steps for creating the user-
defined function are as follows.

1. Clicking the "Tools" menu, then select "Macro", and then select "Visual Basic Editor". The 
Microsoft Visual Basic editor window appears. We will now work in this window (which 
contains several sub windows).

2. On the left is the "Project" sub window. This contains a "project tree" with main headings and 
subheadings. Select "VBAProject (notes09_Theis.xls)". 

3. Click the "Insert" menu and select "Module". A new sub window opens up. The title of this 
sub window is "notes09_Theis.xls - Module1 (Code)". Note that back in the Project sub window, 
an extra item call "Modules" is added under the project "VBAProject (notes09_Theis.xls)".

4. Copy the following code into the module sub window. 

Note that the exponential function is called expint1 instead of E1 because we don't want to fool 
Excel into thinking that we are referring to a cell. Also note that the underscore character ( _ ) is 
the line continuation symbol in Visual Basic. A line that begins with a single quote is a comment 
line.

7. Click the "File" menu and select "Save notes09_Theis.xls". This saves the macro with the 
workbook.

Public Function expint1(u As Double)
    ' Computes the exponential integral E1(u)
    ' Make sure u is not negative
    If u < 0 Then
        expint1 = "#NUM!"
    ElseIf u <= 1 Then
        expint1 = -0.57721566 + 0.99999193 * u _
        - 0.24991055 * u ^ 2 + 0.05519968 * u ^ 3 _
        - 0.00976004 * u ^ 4 + 0.00107857 * u ^ 5 - Log(u)
    Else
        expint1 = Exp(-u) * (u ^ 4 + 8.5773287401 * u ^ 3 _
        + 18.059016973 * u ^ 2 + 8.6347608925 * u + 0.2677737343) _
        / (u ^ 4 + 9.5733223454 * u ^ 3 + 25.6329561486 * u ^ 2 _
        + 21.0996530827 * u + 3.9584969228) / u
    End If
    If expint1 < 1E-100 Then
        ' We avoid returning 0 to prevent triggering
        ' an error message if we plot with log-log axes.
        expint1 = 1E-100
    End If
End Function

To create the interactive Theis type curve, use the following procedure:

1. Put the pumping rate (Q) into cell B2. In this example, we use a pumping rate of 0.00915. This 
is the pumping rate for the data set in Homework #1, problem 2.

2. In column A, starting at row 5, we create a column of values for t/r2. We start at 0.001 and for 
each successive row, we multiply the value in the previous row by 1.2. We proceed down to row 
69, where the value is 116.84.

3. In column B, starting at row 5, we use the Theis equation (3.13) to compute the drawdown for 
the t/r2 value in column A. For example, the formula for cell B5 is

=($B$2/4/PI()/$F$3)*expint1($L$3/4/$F$3/A5)

Note the absolute reference (indicated by the $ symbol) to the cells containing Q, T, and S.

4. Create a log-log plot of drawdown versus time, using data in the cell range A5:B69.

The resultant worksheet should look something like Figure 8.5. When you drag the slider of the 
T or S scrollbar, the computed drawdowns in column B should change, and the type curve should 
also change in the plot.

The behavior of the Theis solution (on log-log plot) can be readily demonstrated by 
varying the T and S values using the scroll bar. As the S value is changed, the type curve is 
translated horizontally to the left or right. As the T value is changed, the type curve is translated 
diagonally. However, the shape of the type curve is always the same.

In another worksheet, we can repeat the above procedure (or use copy and paste) to create 
another Theis type curve on semi-log plot. This is shown in Figure 8.6. The behavior of the Theis 
solution (on semi-log plot) can again be readily demonstrated. As the S value is changed, the 
type curve is translated horizontally to the left or right. As the T value is changed, the slope of 
the straight-line portion is increased or decreased.

Figure 8.5 (on page 8.6). Excel worksheet showing interactive Theis type curve (log-log plot).

Figure 8.6 (on page 8.7). Excel worksheet showing interactive Theis type curve (semi-log plot).

Q T S T t/r2

8.4 Using the Interactive Theis Type Curve for Data Analysis

To illustrate the use of the interactive Theis curve, we analyze the aquifer test data from 
Homework #1 problem 2. We can import the drawdown data from observation wells 1, 2 and 3 
into a separate work sheet, and then compute t/r2 for each data point. Then we can plot the 
observed data on the log-log and semi-log plots with the interactive type curve. It is easy to 
adjust the scrollbars for T and S until the type curve fits the data (to the extent that the type curve 
can fit the data), as illustrated by Fig. 9-7 and 9-8. As you have discovered in Homework #1, the 
drawdown data from observation well 3 do not fall with the data from observation wells 1 and 2 
when all data are plotted on a composite plot (drawdown versus t/r2). In Figures 9-7 and 9-8, the 
type curve is matched primarily to the data from observation wells 1 and 2.

The Excel workbook (notes09_Theis.xls) for this exercise may be downloaded from the 
class web site. 

8.5 Developing an Interactive Type Curve for the Hantush (1960) Solution

An interactive type curve becomes even more useful as the number of aquifer parameter 
increases. In the exercise below, we set up an interactive type curve for the Hantush (1960) 
solution. Please see Section 7.2 of Lecture Notes 7 to review the aquifer setting and assumptions 
of this solution.

The Hantush solution will involve adjustment of 4 parameters. These are:
K = hydraulic conductivity of the pumped aquifer = transmissivity / thickness
Ss = specific storage of the pumped aquifer = storage coefficient / thickness
K’ = hydraulic conductivity of the overlying aquitard
Ss’ = specific storage of the overlying aquitard
Our Excel worksheet will now contain 4 scroll bars, one for each of the above parameters. We 
assume that the aquifer thickness (b) and aquitard thickness (b’) are both known. As in Lecture 
Notes 7, we set the z axis to point downward from the top of the aquitard. Therefore, z = 0 
denotes the top of the aquitard, z = b’ denotes the bottom of the aquitard (also the top of the 
pumped aquifer), and z = b’ + b denotes the bottom of the pumped aquifer.

To evaluate the Hantush solution by the Stehfest algorithm, a user-defined function was 
written using Visual Basic. The function name is hantush1960, and the function parameters are 
as follows:

Hantush1960(t, r, Q, K, Ss, K’, Ss’, b, b’, z)

where t is time, r is radial distance, and Q is pumping rate. The Visual Basic code is shown on 
pages 8.10 and 8.11). 

Figure 9-7. Matching the Theis type curve to observed drawdown on log-log plot.

Figure 9-7. Matching the Theis type curve to observed drawdown on semi-log plot.

.
Public Function Hantush1960(t As Double, r As Double, Q As Double, _
        K As Double, Ss As Double, Kp As Double, Ssp As Double, _
        b As Double, bp As Double, z As Double)
    Dim i As Integer
    Dim p As Double
    Dim sbar As Double
    Dim w As Double  'w here is omega in class notes
    Dim sum As Double
    Dim pi As Double
    Dim v(10) As Double
    v(1) = 8.33333333333333E-02
    v(2) = -32.0833333333333
    v(3) = 1279
    v(4) = -15623.6666666667
    v(5) = 84244.1666666667
    v(6) = -236957.5
    v(7) = 375911.666666667
    v(8) = -340071.666666667
    v(9) = 164062.5
    v(10) = -32812.5
    pi = 3.14159265358979
    sum = 0
    If (z > 0 And z <= bp + b) Then
        For i = 1 To 10
            p = i * Log(2) / t
            w = Sqr(Ssp * p / Kp)
            ' The formula below is eq. (7.35)
            sbar = (Q / 2 / pi / K / b / p) * bessK0(r * _
                Sqr(Ss * p / K + Kp * w * coth(w * bp) / (K * b)))
            ' If computing drawdown in aquitard, use eq (7.32)
            If (z < bp) Then
                sbar = sbar * sr(w * z, w * bp)
            End If
            sum = sum + v(i) * sbar
        Next i
    End If
    ' Make sure the return value is positive so
    ' log-log plot won't complain.
    If sum <= 1E-100 Then
        Hantush1960 = 1E-100
    Else
        Hantush1960 = Log(2) * sum / t
    End If
End Function

Public Function sr(a As Double, b As Double) As Double
    ' This function computes sinh(a)/sinh(b).
    ' Is is assumed that a is always less than b.
    If (b > 100) Then
        sr = Exp(a - b) - Exp(-a - b)
    Else
        sr = (Exp(a) - Exp(-a)) / (Exp(b) - Exp(-b))
    End If
End Function

Public Function coth(x As Double) As Double
    If (x < 100) Then
        coth = (Exp(x) + Exp(-x)) / (Exp(x) - Exp(-x))
    Else
        coth = 1
    End If
End Function

We will use the Hantush (1960) solution to analyze an aquifer test performed at a site 
known simply as Field C. The drawdown data are given by Witherspoon and others (1967, p. 84, 
Table 3-2). The description of Field C is as follows:

• The aquifer extends from 366 m to 411 m below land surface.
• The aquifer is confined above by an aquitard that extends from 244 to 366 m below land 

surface.
• It is assumed that an unpumped aquifer overlies the aquitard, and the pumped aquifer rests on 

an impermeable base.
• The pumping rate was 0.0025 m3/s.
• Observation well 1 is 229 m from the pumped well and  is screened in the pumped aquifer.
• Observation well 2  is 293 m from the pumped well and is screened in the aquitard from 350 

to 351 m below land surface.

For the above setting, we have b = 45 m and b’ = 122 m. For observation well 2, the midpoint of 
the screen is z = 106 m. 

Public Function bessI0(x As Double) As Double
    Dim ax As Double
    Dim t As Double
    ax = Abs(x)
    t = (ax / 3.75)
    If (ax < 3.75) Then
        bessI0 = 1 + 3.5156229 * t ^ 2 + 3.0899424 * t ^ 4 + _
            1.2067492 * t ^ 6 + 0.2659732 * t ^ 8 + 0.0360768 * t ^ 10 _
            + 0.0045813 * t ^ 12
    Else
        bessI0 = (0.39894228 + 0.01328592 / t + 0.00225319 / t ^ 2 _
        - 0.00157565 / t ^ 3 + 0.00916281 / t ^ 4 - 0.02057706 / t ^ 5 _
        + 0.02635537 / t ^ 6 - 0.01647633 / t ^ 7 + 0.00392377 / t ^ 8) _
        * Exp(ax) / Sqr(ax)
    End If
End Function

Public Function bessK0(x As Double) As Double
    Dim y As Double
    If (x < 2) Then
        y = x / 2
        bessK0 = -Log(y) * bessI0(x) - 0.57721566 + 0.4227842 * y ^ 2 + _
            0.23069756 * y ^ 4 + 0.0348859 * y ^ 6 + 0.00262698 * y ^ 8 _
            + 0.0001075 * y ^ 10 + 0.0000074 * y ^ 12
      ElseIf (x < 100) Then
        y = 2 / x
        bessK0 = (1.25331414 - 0.07832358 * y + 0.02189568 * y ^ 2 - _
        0.01062446 * y ^ 3 + 0.00587872 * y ^ 4 - 0.0025154 * y ^ 5 + _
        0.00053208 * y ^ 6) * Exp(-x) / Sqr(x)
      Else
        bessK0 = 0
      End If
End Function

Figure 8.8 and 8.9 show the match of the Hantush (1960) solution to the Field C 
drawdown data respectively on log-log and semi-log plot. The effect of changing K, Ss, K’ or Ss’ 
and be readily seen by use of the scrollbars. 

Figure 9-8. Matching the Hantush (1960) solution to Field C observed drawdown on log-log plot.

Figure 9-8. Matching the Hantush (1960) solution to Field C observed drawdown on semi-log 
plot.


